

Maximális területű négyszög


A négyyszög.mws MAPLE fájlt készítette: dr. Szilassi Lajos

E-mail: szilassi@jgytf.u-szeged.hu

A feladat:

Legyen adott egy négyszög négy oldala. Mutassuk meg, hogy a négyszög területe akkor maximális, ha az húrnégyszög.

Használjuk az ábra jelöléseit:



> **restart;**

Írjuk fel a koszinusztételt az ABC és az ACD háromszögekre:

(Legyen $x = \cos(\beta)$, $y = \cos(\delta)$). Ekkor $\sin(\beta) = (1 - x^2)^{\left(\frac{1}{2}\right)}$ és $\sin(\delta) = (1 - y^2)^{\left(\frac{1}{2}\right)}$.

> **e1:=a^2+b^2-2*a*b*x;**

> **e2:=c^2+d^2-2*c*d*y;**

$$e1 := a^2 + b^2 - 2 a b x$$

$$e2 := c^2 + d^2 - 2 c d y$$

A négyszög területe legyen T .

> **T:=a*b*sqrt(1-x^2)/2 +c*d*sqrt(1-y^2)/2;**

$$T := \frac{a b \sqrt{1 - x^2}}{2} + \frac{c d \sqrt{1 - y^2}}{2}$$

Mivel e -t két háromszög oldalaival és szögeivel is kifejeztük, y , ez alapján T is kifejezhető x -el:

> **y:=solve(e1=e2,y);**

$$y := - \frac{a^2 + b^2 - 2 a b x - c^2 - d^2}{2 c d}$$

> **T;**

$$\frac{a b \sqrt{1 - x^2}}{2} + \frac{c d \sqrt{1 - \frac{(a^2 + b^2 - 2 a b x - c^2 - d^2)^2}{4 c^2 d^2}}}{2}$$

T -t x szerint deriválva keressük T maximumát:

> **g:=diff(T,x);**

$$g := -\frac{a b x}{2 \sqrt{1-x^2}} + \frac{(a^2 + b^2 - 2 a b x - c^2 - d^2) a b}{4 c d \sqrt{1 - \frac{(a^2 + b^2 - 2 a b x - c^2 - d^2)^2}{4 c^2 d^2}}}$$

> **solve(g=0,x);**

$$\frac{-c^2 + a^2 + b^2 - d^2}{2(-c d + a b)}, \frac{-c^2 + a^2 + b^2 - d^2}{2(a b + c d)}$$

> **x1:=1/2*(-c^2+a^2+b^2-d^2)/(a*b+c*d);**

> **x2:= 1/2*(-c^2+a^2+b^2-d^2)/(a*b-c*d);**

$$x1 := \frac{-c^2 + a^2 + b^2 - d^2}{2(a b + c d)}$$

$$x2 := \frac{-c^2 + a^2 + b^2 - d^2}{2(-c d + a b)}$$

Ahhoz, hogy megállapítsuk a szóba jöhető esetek közül melyik a keresett szélsőérték, ideiglenesen adjunk a paramétereknek konkrét értéket:

> **a:=8;b:=7;c:=6;d:=5;**

$$a := 8$$

$$b := 7$$

$$c := 6$$

$$d := 5$$


> **x1;x2;**

$$\frac{13}{43}$$

$$1$$

Rajzoljuk le a T függvényt:

> **plot(T/10,x=-0.2..1);**



Térjünk vissza az általános eset vizsgálatához:

> **a:='a':b:='b':c:='c':d:='d':**

> **x:=1/2*(-c^2+a^2+b^2-d^2)/(a*b+c*d);**

$$x := \frac{-c^2 + a^2 + b^2 - d^2}{2(a b + c d)}$$

> **y;**

$$-\frac{a^2 + b^2 - \frac{a b (-c^2 + a^2 + b^2 - d^2)}{a b + c d} - c^2 - d^2}{2 c d}$$

> **simplify(x+y);**

0

Eszerint $\cos(\beta) = -\cos(\delta)$ és $\sin(\beta) = \sin(\delta)$ azaz β és δ valóban kiegészítő szögek.
Ezt akartuk belátni.

Határozzuk meg a négyszög köréírt körét az előbbi konkrét esetben:

> **k:= r-> arcsin(a/2/r)+ arcsin(b/2/r)+ arcsin(c/2/r)+ arcsin(d/2/r);**

$$k := r \rightarrow \arcsin\left(\frac{1}{2} \frac{a}{r}\right) + \arcsin\left(\frac{1}{2} \frac{b}{r}\right) + \arcsin\left(\frac{1}{2} \frac{c}{r}\right) + \arcsin\left(\frac{1}{2} \frac{d}{r}\right)$$

> **a:=8;b:=7;c:=6;d:=5;**

$a := 8$

$b := 7$

$c := 6$

$d := 5$

> **k(r);**

$$\arcsin\left(\frac{4}{r}\right) + \arcsin\left(\frac{7}{2r}\right) + \arcsin\left(\frac{3}{r}\right) + \arcsin\left(\frac{5}{2r}\right)$$

> **solve(k(r)=Pi);**

$$\frac{\sqrt{15364545}}{840}$$

> **Digits:=15;**

> **fsolve(k(r)=Pi,r=0..8);**

$Digits := 15$

4.66638498299527

> **evalf(1/840*15364545^(1/2));**

4.66638498299527

> **beta:=evalf(180*arccos(x)/Pi);**

$\beta := 72.4026633919202$

> **T;evalf(T);**

$$\frac{\sqrt{1680} \sqrt{1849}}{43}$$

40.9878030638384

Az így kapott eredmények összhangban vannak a [GeoGebrával kapott eredményekkel](#).